


Diode Applications

- ▶ The diode converts AC to DC with a half-wave rectifier.



```
*Half-Wave Rectifier Circuit
.include in4007.mod

Vac vac_in gnd sin (0 169V 60Hz)
;ac line voltage 120vac (rms) is 169Vpp



* transformer
Rs vac_in pri 0.5 ;primary side resistance
Lp pri gnd 10mH ;primary inductance
Ls sec gnd 51uH ;secondary inductance
Kxfrmr Lp Ls 0.999 ;mutual inductance

D1 sec rload in4007
rload rload gnd 1k

.control
set hcopydevtype=postscript
set xbrushwidth=3
set hcopypscolor=0
set color0=rgb:f/f/f
set color1=rgb:0/0/0
set color2=blue
set color3=red


tran 0.25ms 80ms 0ms 0.5ms
plot V(pri) V(sec)
hardcopy half_wave_pri_sec.eps V(pri) V(sec)
tran 0.25ms 30ms 12ms
plot V(sec) V(rload)
hardcopy half_wave_sec_load.eps V(sec) V(rload)
.endc
.end
```

Diode Applications



Diode Applications

- Here is the full-wave bridge rectifier



*Full-Wave Rectifier Circuit

```
.include 1n4007.mod
```

```
Vac vac_in gnd sin (0 169V 60Hz) ;line voltage
```

```
* transformer
```

```
Rs vac_in pri 0.5 ;primary resistance
```

```
Lp pri gnd 10mH ;primary inductance
```

```
Ls sec1 sec2 51uH ;secondary inductance
```

```
Kxfmr Lp Ls 0.999 ;mutual inductance
```

```
* rectifier circuit
```

```
D1 gnd sec1 1n4007
```

```
D2 sec1 rload 1n4007
```

```
D3 sec2 rload 1n4007
```

```
D4 gnd sec2 1n4007
```

```
rload rload gnd 1k
```

```
.control
```

```
set hcopydevtype=postscript
```

```
set xbrushwidth=3
```

```
set hcopypscolor=0
```

```
set color0=rgb:f/f/f
```

```
set color1=rgb:0/0/0
```

```
set color2=blue
```

```
set color3=red
```

```
tran 0.25ms 80ms 0ms 0.5ms
```

```
plot V(pri) V(sec1,sec2)
```

```
hardcopy full_wave_pri_sec.eps V(pri) V(sec1,sec2)
```

```
tran 0.25ms 45ms 12ms
```

```
plot V(sec1,sec2) V(rload)
```



```
hardcopy full_wave_sec_load.eps V(sec1,sec2) V(rload)
```

```
.endc
```

```
.end
```


Diode Applications

► Full-wave rectifier waveforms



Diode Applications

- ▶ Here is a clipper/clamp.



Ngspice simulation of a clipper/clamp


```
.include 1n4148.subckt ;common silicon switching diode
Vcc vcc gnd 5

*initial V, pulsed V, delay, rise, fall, width, period
Vsrc v_in gnd pulse(-10V 10V 0 1m 1m 2.0m 4m)
r1 v_in v_clamped 1k ; current limiting
xd1 v_clamped vcc 1n4148 ; diode
xd2 gnd v_clamped 1n4148 ; diode
r99 v_clamped gnd 1G ; dc path to ground

.control
set hcopydevtype=postscript
set hcopypscolor=0
set xbrushwidth=3
set color0=rgb:f/f/f
set color1=rgb:0/0/0
set color2=blue
set color3=red
* tran step stop start
tran 0.01ms 5ms 0ms
plot V(v_in) V(v_clamped)
hardcopy clipper.eps v(v_in) v(v_clamped)
.endc
.end
```


Diode Applications

- ▶ Here is a junkbox 3V voltage regulator.



tran1: simulation of 3v junkbox voltage regulator

V — v(vout) — v(vin)



simulation of 3V junkbox voltage regulator

```
.include in4007.mod ;common power diode
```

```
*initial V, pulsed V, delay, rise, fall, width, period
```

```
Vsrc vin gnd pulse(0V 6V 0 1m 1m 8.0m 16m)
```

```
r1 vin vout 1k ; current limiting
```

```
d1 vout tie1 in4007 ; diode
```

```
d2 tie1 tie2 in4007 ; diode
```

```
d3 tie2 tie3 in4007 ; diode
```

```
d4 tie3 tie4 in4007 ; diode
```

```
d5 tie4 gnd in4007 ; diode
```

```
.control
```

```
set hcopydevtype=postsript
```

```
set hcopypscolor=0
```

```
set xbrushwidth=3
```

```
set color0=rgb:f/f/f
```

```
set color1=rgb:0/0/0
```

```
set color2=blue
```

```
set color3=red
```

```
* tran step stop start
```

```
tran 0.01ms 3ms 0ms
```

```
plot V(vin) V(vout)
```


```
hardcopy junkbox_reg.eps v(vin) v(vout)
```

```
.endc
```


```
.end
```

Diode Applications

- ▶ Over temperature the junkbox regulator doesn't look so good!



dc2: 3v junkbox voltage regulator over temperature



3V junkbox voltage regulator over temperature


```
.include 1n4007.mod ;common power diode

Vsrc vin gnd 8V ;fixed input voltage
r1 vin vout 1k ; current limiting
d1 vout tie1 1n4007 ; diode
d2 tie1 tie2 1n4007 ; diode
d3 tie2 tie3 1n4007 ; diode
d4 tie3 tie4 1n4007 ; diode
d5 tie4 gnd 1n4007 ; diode


.control
set hcopydevtype=postscript
set hcopypscolor=0
set xbrushwidth=3
set color0=rgb:f/f/f
set color1=rgb:0/0/0
set color2=blue
set color3=red
* temperature start stop step
dc TEMP -50 125 25
plot V(vout)
hardcopy junkbox_reg_temp.eps v(vout)
.endc
.end
```

Diode Applications

- ▶ Here is a peak detector.



tran1: ngspice simulation of a diode peak detector
V — v(out) — v(vac_in)



Ngspice simulation of a diode peak detector


```
.include 1n4148.subckt ;common silicon switching diode

Vac vac_in gnd
sin (0 2V 1000Hz) ;input source at 1KHz, 4vpp
r1 vac_in node1 1k ; current limiting resistor
xd1 node1 out 1n4148 ; diode
c1 out gnd 0.1u ; holding capacitor
r99 out gnd 1G ; required for dc path to ground

.control
set hcopydevtype=postscript
set hcopypscolor=0
set xbrushwidth=2
set color0=rgb:f/f/f
set color1=rgb:0/0/0
set color2=blue
set color3=red
* tran step stop start
tran 100ns 10ms 0ms
plot V(vac_in) V(out)
hardcopy peak_det.eps v(vac_in) v(out)
.endc
.end
```


Diode Applications

- ▶ Here is a AM demodulator.



Ngspice simulation of a diode AM detector


```
.include bat17.subckt ; common silicon switching diode

*Vpeak, Vmin, modulating signal, carrier, offset, phase
Vac vac_in gnd am (2V 0 1kHz 1Meg 0 0) ; am modulated source
r1 vac_in node1 1k ; current limiting resistor
xd1 node1 out bat17 ; schottky diode
c1 out gnd 0.1u ; holding capacitor
r99 out gnd 1000 ; required for dc path to ground


.control
set hcopydevtype=postscript
set hcopypscolor=0
set xbrushwidth=2
set color0=rgb:f/f/f
set color1=rgb:0/0/0
set color2=blue
set color3=red
* tran step stop start
tran 100ns 2ms 0ms
plot V(vac_in) ; big picture
hardcopy am_det_big.eps v(vac_in)
tran 100ns 1.020ms 0.98ms
plot V(vac_in) ; AM waveform zoomed in
hardcopy am_det_zoomout.eps v(vac_in)
tran 100ns 2ms 0ms
plot V(out) ; recovered audio
hardcopy am_det_audio.eps v(out)
.endc
.end
```

Diode Applications


- ▶ Here are the waveforms for the AM demodulator.



Modulated AM Waveform



Close-up, Modulated AM waveform



Recovered Audio Waveform